

Power Switching Semiconductor Models for PSpice, and their Parametric Sensitivity Analysis

X. He¹, B.W. Williams² and T.C. Green²

¹Zhejiang University, Hangzhou, P R China

²Heriot-Watt University, Edinburgh, U. K.

Abstract PSpice models for power semiconductor switching devices, which include the gate turn-off thyristor, the insulated gate bipolar transistor and the mos-controlled thyristor, are presented. These models are suitable for time domain circuit simulation packages which incorporate bipolar transistor and metal oxide semiconductor field effect transistor models. The models presented are composite models, being built-up from these two basic switching components. Simulation, experimentation results and parametric sensitivity analysis validate the in-circuit performance of the proposed composite models. The presented models are meant for applications where circuit level simulated performance is of primary importance, rather than accurate prediction of device microscopic electrical characteristics.

INTRODUCTION

Power semiconductor models have been implemented in general simulation packages [1]-[9] such as PSpice, Saber and so on. Two methods to create a new device model are normally used. The first one is to create a model based on the power semiconductor physical structure and equations, and this kind of model is called the microscopic model. The second modelling method is to construct a model from existing device models, thereby creating a composite model, which gives macroscopic properties. A composite model is simpler to implement and more convenient for power electronics engineers to use in circuit simulation, than the microscopic equation model. This paper surveys power device models, and shows how to construct any new power semiconductor model, with the general circuit simulator PSpice, by composite model building. The insulated gate bipolar transistor (igbt), gate turn-off (gto) thyristor and mosfet controlled thyristor (mct) are presented as three examples of the composite modelling method. Simulations, experimental results and practical device data are given and compared. Parametric sensitivity analysis is performed, and validity of the composite macroscopic models is discussed.

COMPOSITE MODELS

PSpice has models for basic power semiconductor

devices such as diodes, bipolar junction transistors (bjt), and mosfets, which are based on their physical structure and use the equivalent circuit method. But there is no model, in PSpice, for the igbt, gto thyristor or mct, which are newer, more complicated power semiconductor switching devices. Composite models can be created for these devices in PSpice.

Composite igbt model

In view of device operational principles, the igbt is a Darlington combination of an pnp bipolar junction transistor (bjt) and an n-channel mosfet. The igbt equivalent circuit, which is well known, is shown in Fig.1(a), and it combines the input and output characteristics of the mosfet and bjt respectively. The related bjt and mosfet models and their parameter definition can be found in (10)(11).

Table 1 is the result of the parametric sensitivity analysis achieved with the proposed igbt model using the multiple transient analysis capabilities of PSpice. As expected, β_r or τ_r , which are the main model parameters affecting the magnitude and length of the igbt tail current at turn-off, are related to the long average lifetime of holes in wide n-base and the high injection efficiency of the pnp transistor emitter. A large β_r is unwanted in igbts, although it can speed up igbt turn-on, it deteriorates off-state voltage properties. Another factor affecting model characteristics is gate drive. A high performance igbt gate drive circuit is needed to decrease the effect of igbt gate capacitance, C_{gs} , particularly with high voltage and current rated igbts.

Fig.2 shows output $I-V$ characteristics of the igbt composite model used in PSpice. Fig.3 gives a comparison of the igbt transfer characteristics between the model and practical device; BSM50GB160D (12) or similar (13)(14). Fig.4 shows igbt circuit simulation and experimental results, at turn-off. It is seen in these dynamic waveforms, that tail current properties are an inherent feature of the igbt model (15). By adjustment of β_r and τ_r the tail current magnitude and its length can be varied to produce igbt characteristics adopted by other manufactures.




Fig.1 The composite models of the IGBT (left), GTO (middle) and MCT (right).

Table 1: IGBT model parameter sensitivity analysis results

Δ	Values	Turn-on			Turn-off					Steady state	
		t_d	t_r	P_{ton}	t_{fd}	t_r	t_{tail}	I_{tail}	P_{toff}	V_{on}	I_{ces}
I_{sM} (nA)	20.0	-	-	-	-	↑	-	↑	↑	-	Δ
R_g (k Ω)	1.0	Δ	Δ	Δ	Δ	↑	-	-	↑	-	-
C_{gs} (nF)	0.2	Δ	Δ	Δ	Δ	↑	↑	-	↑	-	-
R_d (Ω)	0.02	-	-	-	-	-	-	-	-	Δ	-
C_{gd} (pF)	1.0	↑	Δ	Δ	↑	-	-	-	↑	-	-
K_p	7.6	↓	↓	↓	-	↑	↓	-	-	∇	-
V_{TO} (V)	6.0	↑	Δ	Δ	↓	∇	↑	-	↓	Δ	-
I_{sT} (nA)	4.0	-	-	-	-	-	-	-	-	-	Δ
R_c (Ω)	0.03	-	-	-	-	-	-	-	-	Δ	-
C_{je} (nF)	0.6	-	-	-	↑	↓	-	-	-	-	-
β_f	0.26	-	-	-	↑	↑	Δ	Δ	Δ	∇	↑
τ_f (μ s)	3.3	-	↑	↑	-	-	Δ	-	Δ	↑	-
β_r	0.44	-	-	-	-	-	-	-	-	-	↓
T ($^{\circ}$ C)	25.0	↑	↑	Δ	↓	↑	-	-	↑	Δ	↑

Note: Δ ∇ significant effect, \uparrow \downarrow small effect, -- no effect.




Fig.2 IGBT model output curve. $V_o=V_{ce}$, $I_E(X1,Q1)=I_c$.




Fig.3 IGBT transfer characteristic. $T_j < 120^{\circ}$ C.

Table 2: GTO model parameter sensitivity analysis results

Δ	Values	Turn-on			Turn-off					Steady state				
		t_d	t_r	P_{ion}	t_{fd}	t_f	t_{tail}	I_{tail}	P_{toff}	V_{on}	I_{ces}	I_L	I_H	
I_{SP} (nA)	1.0	-	-	-	-	-	-	-	-	-	-	↑	-	-
β_{FP}	0.33	-	-	-	Δ	-	↑	↑	Δ	↓	↑	▽	▽	
β_{RP}	1.2	-	-	-	Δ	↓	↓	-	-	-	↓	-	-	
R_{CP} (Ω)	0.05	-	-	-	-	-	-	-	-	Δ	-	-	-	
τ_{FP} (μ s)	14.0	-	-	-	-	-	Δ	-	Δ	-	-	Δ	-	
τ_{RP} (μ s)	0.2	-	-	-	Δ	▽	-	-	↑	-	-	-	-	
I_{SN} (nA)	15.0	-	-	-	-	-	-	-	-	↓	↑	▽	▽	
β_{FN}	110.0	-	-	↓	↑	↓	-	-	↓	↓	↑	▽	▽	
β_{RN}	0.08	-	-	-	-	-	-	-	-	-	↓	-	-	
R_{DN} (Ω)	0.05	-	-	↑	-	-	-	-	-	↑	-	↑	↑	
R_{CN} (m Ω)	4.2	-	↓	↓	-	-	-	-	-	Δ	-	-	-	
C_{jCN} (nF)	2.2	-	-	-	-	-	-	-	-	-	-	-	-	
C_{jeN} (nF)	0.2	-	-	-	↑	↓	-	-	-	-	-	-	-	
τ_{FN} (μ s)	0.28	↑	-	↑	↑	-	-	-	-	-	-	Δ	-	
Area	1.0	-	↑	↑	-	↑	-	-	↑	↓	Δ	↓	▽	
T ($^{\circ}$ C)	27.0	-	-	↓	↓	↑	-	-	↑	↓	Δ	▽	↓	

Note: Δ ∇ significant effect, \uparrow \downarrow small effect, -- no effect.

Subscripts: P refers to the p-n-p transistor, and N refers to the n-p-n transistor, of the GTO model.

Table 3. Basic GTO model and device characteristics. ($T_j=125^{\circ}$ C unless otherwise stated)

	Definition	DGT224SE(max.)	Model	Simulation Conditions
V_{on} (V)	On-state voltage.	2.3	2.29	$I_T=401(A)$, $I_G(on)=2.0(A)$.
I_{ces} (mA)	Off-state current.	25	23.0	$V_{DC}=1300(V)$, $V_{RG}=2(V)$.
V_{GT} (V)	Gate trigger voltage.	0.9	0.89	$V_{DC}=24(V)$, $I_T=103(A)$,
I_{GT} (A)	Gate trigger current.	1	0.54	$T_j=25^{\circ}C$.
t_d (μ s)		1.5	1.0	$V_{DC}=750(V)$, $I_T=415(A)$,
t_r (μ s)		3	3.7	$I_{FG}=10(A)$, rise $t_{FG}=0.9\mu$ s,
t_{on} (μ s)	$=t_d+t_r$. Turn-on time.	4.5	4.7	Resistive load.
E_{ON} (mJ)	Turn-on energy.	135	157.0	
t_{fd} (μ s)		7	6.0	$V_{DC}=750(V)$, $I_T=400(A)$,
t_f (μ s)		1.3	2.2	Snub. $C_s=1(\mu$ F),
t_{off} (μ s)	$=t_{fd}+t_f$. Turn-off time.	8.3	8.2	$di_{GQ}/dt=18(A/\mu$ s),
t_{tail} (μ s)			10.3	Resistive load.
E_{OFF} (mJ)	Turn-off energy from $0.9I_T$ - $0.1I_T$.	250	222.4	

Note: The model parameter values are given in table 2.




Fig.4 igbt turn-off simulated (top) and experimental (bottom). (igbt: BSM50GB100D).

Composite gto thyristor model

The composite gto thyristor model is derived by regeneratively connecting pnp and npn bjt models, as shown in Fig.1(b). The model was first used to explain how a thyristor operates (16), and then how a gto device operates. The model's parametric sensitivity analysis in table 2 demonstrates the relationship between the model parameters and device electrical characteristics, and can be used as a basis to simulate different gto devices (7). The various parameter definitions can be found in (7)(11).

The practical device selected to illustrate the composite model is the DGT224SE (17) or similar (18)(19). Table 3 gives a comparison of the gto device and model performance, and the model parameter values used are listed in table 2. Device turn-off ability depends on the capacitance of the snubber capacitor C_s . Fig.5 shows the relationship between maximum controllable current and C_s for the device and its PSpice model. The composite gto model inherently models gto tail current characteristics. Fig.6 is gto anode current and voltage waveforms at gto turn-off, with different β 's for the pnp bjt. The model can be matched to different practical gto devices by changing the appropriate parameter values given in table 3.

Composite mct model

The composite model concept is applied to analyze an mct, as a mosfet driving the gate of a gto thyristor.

Fig.1(c) shows the mct model equivalent circuit, which is based on the Harris Semiconductor model (9). The parametric sensitivity analysis for the composite mct model is given in table 4. It is important that the bjt's realistically model gto features.

A composite mct model can also be configured by a mosfet driving the gate of a gto which consists of three diodes and current source, rather than pnp and npn bipolar transistors (20). More components, such as resistors and interactive switches, are needed to obtain simulation results which agree with the practical mct.

DISCUSSION

The composite power semiconductor models, based on the bjt, mosfet, diode and resistor models which are incorporated in most circuit simulators, are simple to configure, easy to use and possess the features of practical devices. They are a good compromise between computational time and accuracy. The microscopic models, such as the gto 1 dimension (1-D), 2 dimensions (2-D) or partly 2-D models, are too complicated and time consuming when it is circuit and system performance that are of prime interest. A composite model is circuit or system oriented, but based on device physics analysis. This model concept is useful for constructing, analysing and developing new power electronics devices, such as the double gate gto (21) where the device structure should be considered for the composite gto model.

The complete PSpice file for the igbt, gto and mct composite models are given in the appendix, including circuit component values. The proposed models are constructed in PSpice, but the composite model method, as a general approach, is suitable for use in other simulation packages.

CONCLUSION

The composite modelling method for power switching semiconductor device modelling is simple and a useful way for power electronics engineers to perform circuit and system simulations. Each proposed power semiconductor model has the inherent controllable and adjustable static, dynamic and thermal characteristics of the practical device.

The paper described three composite power semiconductor device (igbt, gto and mct) models, and gives model parametric sensitivity analysis. The models make a good compromise between model accuracy and computational time. Included are the PSpice files for each model, such that any PSpice user can readily incorporate and use realistic and simple models for the igbt, gto thyristor and mct.




Fig.5 Dependence of maximum controllable current I_{TCM} on capacitor C_s .

$V_{DC}=750(V)$, $L_p=0.8(\mu H)$, $L_s=2(\mu H)$, $R_s=4(\Omega)$.




Fig.6 GTO voltages (top) and currents (bottom) at turn-off, with different gain β of the p-n-p BJT in the GTO composite model.

Table 4: The composite MCT model parameter sensitivity analysis

Δ	Values	Turn-on			Turn-off					Steady state		
		t_d	t_r	P_{ton}	t_{fd}	t_f	t_{tail}	I_{tail}	P_{toff}	V_{on}	I_{ces}	I_H
K_{nF}	2.4	-	↓	-	-	-	-	↑	↑	-	-	-
V_{nTO} (V)	5.0	-	-	-	-	-	-	-	-	-	-	-
K_{pF}	3.6	-	-	-	-	-	↑	↓	-	↓	-	↑
V_{pTO} (V)	-3.0	-	-	-	-	↑	↑	↓	-	-	-	-
β_{nF}	0.4	-	-	-	↑	↑	Δ	Δ	Δ	▽	↑	▽
β_{nR}	1.0	-	-	-	-	-	-	-	-	↑	↓	-
τ_{nF} (μs)	1.2	-	-	-	-	-	↑	-	↑	-	↓	↓
τ_{nR} (μs)	1.0	-	-	-	-	-	-	-	-	-	-	-
β_{pF}	2.4	-	-	-	↑	-	Δ	Δ	Δ	▽	-	▽
β_{pR}	1.0	-	-	-	-	-	-	-	-	-	-	-
τ_{pF} (μs)	1.2	-	-	-	-	-	Δ	-	Δ	-	-	↑
τ_{pR} (μs)	1.0	-	-	-	-	-	-	-	-	-	-	-
T (°C)	27.0	-	-	↑	-	-	Δ	↑	Δ	-	↑	▽

Note: Δ▽ significant effect, ↑↓ small effect, -- no effect.

REFERENCES

- (1) P.O. Lauritzen, 'Power Semiconductor Device Model for Use in Circuit Simulators', IEEE IAS Annual Meeting Record, October 1990, pp.1559--1563, Seattle, USA
- (2) M. Andersson, P. Kuivalainen, and H. Pohjonen, 'Circuit Simulation Models for MOS-Gated Power Devices: Application to The Simulation of An Electronic Lamp Ballast Circuit', IEEE APEC Record, March 1993, pp.498--503, San Diego, USA.
- (3) F.F. Protiwa, O. Apeldoorn, N. Groos, 'New IGBT Model for PSpice', EPE'93 Conf. Record, September 1993, Vol.2, pp.226--231, Brighton, UK
- (4) C.L. Tsay, R. Fischl, J. Schwartzberg, H. Kan and J. Barrow, 'A High Power Circuit Model for the Gate Turn off Thyristor', IEEE PESC Record, June 1990, pp.390--397, San Antonio, USA
- (5) O. Apeldoorn, L. Schulting, and H.-Ch. Skudelnny, 'Modelling A GTO with PSpice', IEEE IAS Annual Meeting Record, October 1992, pp.1074--1081, Michigan, USA
- (6) P.R. Palmer and K.J. Tseng, 'An Accurate GTO Model for Circuit Simulations', EPE'93 Conf. Record, September 1993, Vol.2, pp.244--248, Brighton, UK
- (7) X. He, B.W. Williams and T.C. Green, 'A Composite Gate Turn-off Thyristor Model And Its Application'. IEEE IAS Annual Meeting Record, October 1993, pp.1202--1207, Toronto, Canada
- (8) S.K. Sul, F. Profumo, C.H. Cho and T.A. Lipo, 'MCTs and IGBTs: A Comparison of Performances in Power Electronics Circuits', IEEE PESC Record, June 1989, pp.163--169, Wisconsin, USA
- (9) _____, 'MOS Controlled Thyristor-user's guide', HARRIS Semiconductors, 1993, USA
- (10) P. Antogretti and G. Massobrio, 'Semiconductor Device Modelling with SPICE', McGraw-Hill, 1988
- (11) _____, 'PSpice Manual', MicroSim Corporation, July 1991, USA
- (12) _____, 'IGBT Module Data Sheets', Siemens AG, Semiconductors group, June 1993, Germany (13) _____, "'S" Series MOSIGBTs', Data book No.91560B, IXYS Corporation. July 1992, USA
- (14) _____, 'Silicon N-Channel IGBT', Product Preview, Hitachi Co., July 1992, Japan
- (15) X. He, S.J. Finney, B.W. Williams and T.C. Green, 'Improved Passive Lossless Turn-on and Turn-off Snubber', IEEE APEC Record, March 1993, pp.385--392, San Diego, USA
- (16) B.W. Williams, 'Determination of Power Semiconductor Model Parameter Values from Structure Data', Solid-State Electronics, Vol.25, No.5, pp.395--410, 1982, U.K.
- (17) _____, 'Gate Turn-Off Thyristor Manual', GEC Plessey Semiconductors, October 1992, UK
- (18) _____, 'Gate Turn-Off Thyristor', AEG Power Semiconductors Ltd. 1986, UK
- (19) _____, 'WG Series of GTO Thyristors. -Notes on Ratings and Characteristics', Westcode Semiconductors Ltd., October 1992, England
- (20) S. Yuvarajan, D. Quek and J. Weimer, 'Switching Characteristics And PSpice Model of An MCT', IEEE IAS Annual Meeting Record, October 1993, pp.1208--1215, Toronto, Canada
- (21) T. Ogura et al. 'High Frequency 6000V Double Gate GTOs with Buried Gate Structure', Proceeding of Inter. Symposium on Power Semiconductor Devices & ICs, 1990, pp.252--255, Tokyo, Japan

APPENDIX

Simulated circuit

The simulated circuit for the igbt is given in Fig.7, and can be used for the gto thyristor and mct circuit simulation by changing the circuit component values.




Fig.7 Simulated circuit.

IGBT model and circuit file

```

Vo 1 0 DC 600.0
RI 1 2 12.0
L1 2 3 300U IC=40
Df 3 1 Dfast
Ls 3 4 20U IC=40
Ds 4 8 Dfast
Cs 8 0 0.047U IC=0
Rs 8 3 20
Vg 7 0 PULSE(15 -15 2U 100N 100N 50U 100U)

```

```

Dgon 7 11 Dnorm
Rgon 11 6 47
Dgof 12 7 Dnorm
Rgof 12 6 3.3
X1 4 6 0 IGBT
.SUBCKT IGBT 4 6 10
M1 5 6 10 10 MOSN
Q1 10 5 4 BJTP
.ENDS
.MODEL MOSN NMOS(IS=20N KP=7.6 VTO=6V
+ RD=0.02 RG=1000 CGSO=2U CGDO=10N)
.MODEL BJTP PNP(BF=0.26 TF=3.3U CJC=0.6N
+ RE=0.03 BR=0.44 IS=4N)
.MODEL Dfast D()
.MODEL Dnorm D()
.TEMP 25
.TRAN 0.4U 180.0U 40U 0.6U UIC
.PROBE
.OPTIONS (ITL4=40 ITL5=15000 ABSTOL=100.0N
+ VNTOL=1.0M LIMPTS=10000)
.END

```

GTO model and circuit file

```

Vo 1 0 DC 600
R1 1 2 1.0
L1 2 3 4M IC=400
Df 3 1 Dfast
Ls 3 4 12U IC=400
Cs 88 0 4U IC=0
Ds 4 8 Dfast
Lo 8 88 0.8U IC=0
Rs 8 3 4
Vg 7 0 PULSE(5 -5 10U 1U 2U 105U 200U)
Rg 7 12 2
Vgon 9 0 PULSE(-25 10.0 110U 4U 4U 10U 200U)
Dg1 9 13 Dnorm
Rg1 13 12 0.3
Vgoff 11 0 PULSE(15 -20 10U 12U 40U 8U 200U)
Dg2 12 11 Dfast
Rg2 12 6 0.14
.SUBCKT GTO 4 6 10
Qpnp 6 5 4 BJTP
Qnpn 5 6 10 BJTN
.ENDS
Xsw 4 6 0 GTO
.MODEL Dnorm D()
.MODEL Dfast D()
.MODEL BJTP PNP(BF=0.33 TF=14U TR=0.2U
+ RC=0.05 BR=1.2 IS=1N)
.MODEL BJTN NPN(BF=110 TF=0.28U CJE=0.5N
+ RB=0.05 IS=15N CJC=2.2N MJC=0.5
+ RC=0.0042 BR=0.08)
.TRAN 0.2U 480U 40U 0.5U UIC
.PROBE
.OPTIONS (ITL4=40 ITL5=18000 ABSTOL=200.0U
+ VNTOL=20.0M LIMPTS=20000)
.END

```

MCT model (p channel) and circuit file

```

Vo 1 0 DC 450
R1 1 2 2.6
L1 2 3 8M IC=150
Df 3 1 Dfast
Ls 3 4 4U IC=0
Cs 8 0 1U IC=450
Ds 4 8 Dfast
Rs 8 3 4
Vg 4 21 PULSE(-10 20 2U 400N 800N 40U 80U)
.SUBCKT MCTS 4 21 10
Mon 6 21 4 4 MOSP
Qpnp 6 5 4 BJTP
Moff 5 21 4 4 MOSN
Qnpn 5 6 10 BJTN
.ENDS
Xsw 4 21 0 MCTS
.MODEL Dfast D()
.MODEL MOSN NMOS(KP=2.4 VTO=5)
.MODEL MOSP PMOS(KP=3.6 VTO=-3)
.MODEL BJTP PNP(BF=2.4 TF=1.2U TR=1U)
.MODEL BJTN NPN(BF=0.4 TF=1.2U TR=1U)
.TRAN 0.4U 160U 0U 1000N UIC
.PROBE
.OPTIONS (ITL4=40 ITL5=15000 ABSTOL=100.0N
+ VNTOL=100.0U LIMPTS=15000)
.END

```